# Maintainer: Caleb Maclennan <caleb@alerque.com>
# Contributor: Simo Leone <simo@archlinux.org>
# Contributor: Rouslan Solomakhin <rouslan@localnet.com>

pkgname=aspell-ru
_pkgver=0.99f7-1
pkgver=${_pkgver//-/.}
pkgrel=1
pkgdesc="Russian dictionary for aspell"
arch=('loong64' 'x86_64') # We cannot use 'any' see FS#22443
url='http://aspell.net'
license=(custom)
depends=(aspell)
_archive="aspell6-ru-$_pkgver"
source=("https://ftp.gnu.org/gnu/aspell/dict/ru/$_archive.tar.bz2")
sha256sums=('5c29b6ccce57bc3f7c4fb0510d330446b9c769e59c92bdfede27333808b6e646')

build() {
	cd "$_archive"
	./configure
	make
}

package() {
	cd "$_archive"
	make DESTDIR="$pkgdir" install
	install -Dm0644 -t "$pkgdir/usr/share/licenses/$pkgname/" Copyright
}
